

The Stochastic Power Law Logistic Model: Derivations of Asymptotic Approximations of the first 3 Cumulants of the QSD, with $s=5$.

Ingemar Nåsell

The procedure **cumf** is used to determine the cumulant of order kk as a function of the (raw) moments of orders up to kk , while the procedure **muf** determines the (raw) moment of order kk as a function of the cumulants of orders up to kk . The procedure **ODEcum3** derives ODEs for the first 3 cumulants.

More details are given in the Maple work-sheet DeriveODECum3.

```
> restart;
> cumf:=proc(kk)
    local M,K,Ks,cum;
    description "Determines the cumulant of order kk as function of
the raw moments of orders up to kk";
    M:=1+add(cat(mu,k)*theta^k/k!,k=1..kk);
    K:=log(M);
    Ks:=convert(series(K,theta,kk+1),polynom);
    cum:=sort(simplify(coeff(Ks,theta,kk)*kk!),[seq(cat(mu,j),j=1..
kk)],plex);
end proc;

> muf:=proc(kk)
    local K,M,Ms,mu;
    description "Determines the (raw) moment of order kk as function
of the cumulants of orders up to kk";
    K:=add(cat(kappa,k)*theta^k/k!,k=1..kk);
    M:=exp(K);
    Ms:=convert(series(M,theta,kk+1),polynom);
    mu:=sort(coeff(Ms,theta,kk)*kk!,[seq(cat(kappa,j),j=1..kk)],
plex);
end proc;

> ODEcum3:=proc(s)
    local Dmu1,Dmu2,Dmu3,sub1,sub2,Dcum1,Dcum2,Dcum3,a,c,d,b;
    description "Derive ODEs for the first 3 cumulants";
    Dmu1:=a*mu1-b*cat(mu,s+1);
    Dmu2:=2*a*mu2+c*mu1-d*cat(mu,s+1)-2*b*cat(mu,s+2);
    Dmu3:=a*(mu1+3*mu3)+3*c*mu2-3*d*cat(mu,s+2)-b*(cat(mu,s+1)+3*
cat(mu,s+3));
    sub1:=D(mu1)=Dmu1,D(mu2)=Dmu2,D(mu3)=Dmu3;
    sub2:=seq(cat(mu,i)=muf(i),i=1..3+s);
    Dcum1:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(1))))),[a,c,d,b])));
```

```

Dcum2:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(2)))),[a,c,d,b])));
Dcum3:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(3)))),[a,c,d,b])));
a:=mu*(R0-1);
c:=mu*(R0+1);
d:=mu*(R0-alpha)/N^s;
b:=mu*(R0+alpha)/N^s;
[eval(Dcum1),eval(Dcum2),eval(Dcum3)];
end proc:

```

Put $s=5$ and denote the derivatives of the first 3 cumulants by A, B, C.

```
> s:=5;
```

```
s := 5
```

(1)

```

> A:=ODEcum3(s)[1];
B:=ODEcum3(s)[2];
C:=ODEcum3(s)[3];

```

$$A := \mu (R0 - 1) \kappa 1 - \frac{1}{N^5} \left((\kappa 1^6 + 15 \kappa 1^4 \kappa 2 + 20 \kappa 1^3 \kappa 3 + 15 \kappa 2^3 + (45 \kappa 2^2 + 15 \kappa 4) \kappa 1^2 + 15 \kappa 2 \kappa 4 + 10 \kappa 3^2 + (60 \kappa 2 \kappa 3 + 6 \kappa 5) \kappa 1 + \kappa 6) \mu (R0 + \alpha) \right)$$

$$B := 2 \mu (R0 - 1) \kappa 2 + \mu (R0 + 1) \kappa 1 - \frac{1}{N^5} \left((\kappa 1^6 + 15 \kappa 1^4 \kappa 2 + 20 \kappa 1^3 \kappa 3 + 15 \kappa 2^3 + (45 \kappa 2^2 + 15 \kappa 4) \kappa 1^2 + 15 \kappa 2 \kappa 4 + 10 \kappa 3^2 + (60 \kappa 2 \kappa 3 + 6 \kappa 5) \kappa 1 + \kappa 6) \mu (R0 - \alpha) \right) - \frac{1}{N^5} \left(2 (6 \kappa 1^5 \kappa 2 + 15 \kappa 1^4 \kappa 3 + 60 \kappa 1^3 \kappa 2^2 + 20 \kappa 1^3 \kappa 4 + 150 \kappa 1^2 \kappa 2 \kappa 3 + 90 \kappa 1 \kappa 2^3 + 15 \kappa 1^2 \kappa 5 + 90 \kappa 1 \kappa 2 \kappa 4 + 60 \kappa 1 \kappa 3^2 + 105 \kappa 2^2 \kappa 3 + 6 \kappa 1 \kappa 6 + 21 \kappa 2 \kappa 5 + 35 \kappa 3 \kappa 4 + \kappa 7) \mu (R0 + \alpha) \right)$$

$$C := (\kappa 1 + 3 \kappa 3) \mu (R0 - 1) + 3 \mu (R0 + 1) \kappa 2 - \frac{1}{N^5} \left(3 (6 \kappa 1^5 \kappa 2 + 15 \kappa 1^4 \kappa 3 + 60 \kappa 1^3 \kappa 2^2 + 20 \kappa 1^3 \kappa 4 + 150 \kappa 1^2 \kappa 2 \kappa 3 + 90 \kappa 1 \kappa 2^3 + 15 \kappa 1^2 \kappa 5 + 90 \kappa 1 \kappa 2 \kappa 4 + 60 \kappa 1 \kappa 3^2 + 105 \kappa 2^2 \kappa 3 + 6 \kappa 1 \kappa 6 + 21 \kappa 2 \kappa 5 + 35 \kappa 3 \kappa 4 + \kappa 7) \mu (R0 - \alpha) \right) - \frac{1}{N^5} \left((\kappa 1^6 + 18 \kappa 1^5 \kappa 3 + (90 \kappa 2^2 + 15 \kappa 2 + 45 \kappa 4) \kappa 1^4 + 270 \kappa 2^4 + (540 \kappa 2 \kappa 3 + 20 \kappa 3 + 60 \kappa 5) \kappa 1^3 + 15 \kappa 2^3 + 585 \kappa 2^2 \kappa 4 + (540 \kappa 2^3 + 45 \kappa 2^2 + 630 \kappa 2 \kappa 4 + 450 \kappa 3^2 + 15 \kappa 4 + 45 \kappa 6) \kappa 1^2 + 10 \kappa 3^2 + 168 \kappa 3 \kappa 5 + 105 \kappa 4^2 + (1710 \kappa 2^2 \kappa 3 + 630 \kappa 3 \kappa 4 + (60 \kappa 3 + 360 \kappa 5) \kappa 2 + 6 \kappa 5 + 18 \kappa 7) \kappa 1 + (810 \kappa 3^2 + 15 \kappa 4 + 81 \kappa 6) \kappa 2 + \kappa 6 + 3 \kappa 8) \mu (R0 + \alpha) \right)$$

(2)

Determine asymptotic approximations of the first 3 cumulants of the following forms. Assume also that $\kappa 4 - \kappa 8$ are $O(N)$.

```
> kappal:=x1*N + x2 + x3/N;
```

```

kappa2:=y1*N + y2:
kappa3:=z1*N:
kappa4:=u1*N:
kappa5:=u2*N:
kappa6:=u3*N:
kappa7:=u4*N:
kappa8:=u5*N:

```

Asymptotic expressions for A, B, C are written $A=A1*N + A2 + A3/N$, $B=B1*N + B2$, $C=C1*N$, where

```

> A1:=coeff(A,N,1); A2:=coeff(A,N,0); A3:=coeff(A,N,-1);
B1:=coeff(B,N,1); B2:=coeff(B,N,0);
C1:=coeff(C,N,1);

```

$$A1 := \mu (R0 - 1) x1 - x1^6 \mu (R0 + \alpha)$$

$$A2 := \mu (R0 - 1) x2 - (6 x2 x1^5 + 15 x1^4 y1) \mu (R0 + \alpha)$$

$$A3 := \mu (R0 - 1) x3 - (60 x2 x1^3 y1 + 15 x1^4 y2 + 20 x1^3 z1 + 45 x1^2 y1^2 + 2 (x3 x1^2 + 2 x2^2 x1 + x1 (2 x3 x1 + x2^2)) x1^3 + 9 x2^2 x1^4) \mu (R0 + \alpha)$$

$$B1 := 2 \mu (R0 - 1) y1 + \mu (R0 + 1) x1 - x1^6 \mu (R0 - \alpha) - 12 x1^5 y1 \mu (R0 + \alpha)$$

$$B2 := 2 \mu (R0 - 1) y2 + \mu (R0 + 1) x2 - (6 x2 x1^5 + 15 x1^4 y1) \mu (R0 - \alpha) - 2 (6 x1^5 y2 + 30 x2 x1^4 y1 + 15 x1^4 z1 + 60 x1^3 y1^2) \mu (R0 + \alpha)$$

$$C1 := (x1 + 3 z1) \mu (R0 - 1) + 3 \mu (R0 + 1) y1 - 18 x1^5 y1 \mu (R0 - \alpha) - (x1^6 + 18 x1^5 z1 + 90 y1^2 x1^4) \mu (R0 + \alpha) \quad (3)$$

6 equations can now be formed by setting each of A1, A2, A3, B1, B2, C1 equal to zero. These equations can be solved for the 6 unknowns x1, x2, x3, y1, y2, z1. The solutions are found sequentially, as follows:

First we solve $A1=0$ for $x1$. This equation has $s+1=6$ solutions. Among them, we exclude $s=5$ spurious solution.

After this, $B1=0$ is solved for $y1$, $A2=0$ is solved for $x2$, $C1=0$ is solved for $z1$, $B2=0$ is solved for $y2$, and $A3=0$ is solved for $x3$.

```

> x1a:=solve(A1,x1);
x1:=((R0-1)/(R0+alpha))^(1/5);

```

$$x1a := 0, \frac{((R0 - 1) (R0 + \alpha)^4)^{1/5}}{R0 + \alpha},$$

$$\frac{\left(-\frac{1}{4} + \frac{\sqrt{5}}{4} + \frac{I\sqrt{2}\sqrt{5+\sqrt{5}}}{4}\right) ((R0 - 1) (R0 + \alpha)^4)^{1/5}}{R0 + \alpha},$$

$$\frac{\left(-\frac{1}{4} - \frac{\sqrt{5}}{4} + \frac{I\sqrt{2}\sqrt{5-\sqrt{5}}}{4}\right) ((R0 - 1) (R0 + \alpha)^4)^{1/5}}{R0 + \alpha},$$

$$\frac{\left(-\frac{1}{4} - \frac{\sqrt{5}}{4} - \frac{I\sqrt{2}\sqrt{5-\sqrt{5}}}{4}\right) \left((R0-1)(R0+\alpha)^4\right)^{1/5}}{R0+\alpha},$$

$$\frac{\left(-\frac{1}{4} + \frac{\sqrt{5}}{4} - \frac{I\sqrt{2}\sqrt{5+\sqrt{5}}}{4}\right) \left((R0-1)(R0+\alpha)^4\right)^{1/5}}{R0+\alpha}$$

$$x1 := \left(\frac{R0-1}{R0+\alpha}\right)^{1/5} \quad (4)$$

> `y1:=factor(solve(B1,y1));`

$$y1 := \frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/5} R0 (\alpha+1)}{5 (R0+\alpha) (R0-1)} \quad (5)$$

> `x2:=solve(A2,x2);`

$$x2 := -\frac{3 R0 (\alpha+1)}{5 (R0+\alpha) (R0-1)} \quad (6)$$

> `z1:=factor(solve(C1,z1));`

$$z1 := -\frac{R0 \left(\frac{R0-1}{R0+\alpha}\right)^{1/5} (\alpha+1) (5 R0^2 - R0\alpha - R0 + 5\alpha)}{25 (R0+\alpha)^2 (R0-1)^2} \quad (7)$$

> `y2:=solve(B2,y2);`

$$y2 := \frac{3 (R0^2 + \alpha) R0 (\alpha+1)}{5 (R0+\alpha)^2 (R0-1)^2} \quad (8)$$

> `x3:=factor(solve(A3,x3));`

$$x3 := -\frac{(\alpha+1) (25 R0^2 + 4 R0\alpha + 4 R0 + 25\alpha) \left(\frac{R0-1}{R0+\alpha}\right)^{4/5} R0}{25 (R0-1)^3 (R0+\alpha)} \quad (9)$$

Summarize the results for s=5:

> `x1, x2, x3, y1, y2, z1;`

$$\left(\frac{R0-1}{R0+\alpha}\right)^{1/5}, -\frac{3 R0 (\alpha+1)}{5 (R0+\alpha) (R0-1)}, \quad (10)$$

$$-\frac{(\alpha+1) (25 R0^2 + 4 R0\alpha + 4 R0 + 25\alpha) \left(\frac{R0-1}{R0+\alpha}\right)^{4/5} R0}{25 (R0-1)^3 (R0+\alpha)},$$

$$\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/5} R0 (\alpha+1)}{5 (R0+\alpha) (R0-1)}, \frac{3 (R0^2 + \alpha) R0 (\alpha+1)}{5 (R0+\alpha)^2 (R0-1)^2},$$

$$\frac{RO \left(\frac{RO-1}{RO+\alpha} \right)^{1/5} (\alpha+1) (5RO^2 - RO\alpha - RO + 5\alpha)}{25 (RO+\alpha)^2 (RO-1)^2}$$

